Predictive Modeling and Analysis

Business Analytics, 1st edition
James R. Evans

- Logic-Driven Modeling
- Data-Driven Modeling
- Analyzing Uncertainty and Model Assumptions
- Model Analysis Using *Risk Solver Platform*

Logic-Driven Modeling

- Predictive modeling is the heart and soul of business decisions.
- Building decision models is more of an art than a science.
- Creating good decision models requires:
  - solid understanding of business functional areas
  - knowledge of business practice and research
  - logical skills
- It is best to start simple and enrich models as necessary.

Logic-Driven Modeling

Example 8.1 The Economic Value of a Customer

- A restaurant customer dines 6 times a year and spends an average of $50 per visit.
- The restaurant realizes a 40% margin on the average bill for food and drinks.
- Annual gross profit on a customer = $50(6)(0.40)
  = $120
- 30% of customers do not return each year.
- Average lifetime of a customer = 1/.3 = 3.33 years
- Average gross profit for a customer = $120(3.33)
  = $400

Logic-Driven Modeling

Example 8.1 (continued)
The Economic Value of a Customer

$$V = \frac{R \times F \times M}{D}$$

- V = value of a loyal customer
- R = revenue per purchase
- F = purchase frequency (number visits per year)
- M = gross profit margin
- D = defection rate (proportion customers not returning each year)

Logic-Driven Modeling

Example 8.2 A Profit Model

- Develop a decision model for predicting profit in face of uncertain demand.

P = profit
R = revenue
C = cost

p = unit price
c = unit cost
F = fixed cost
S = quantity sold
D = demand
Q = quantity produced



Figure 8.1

Logic-Driven Modeling

Example 8.2 (continued) A Profit Model

- Cost = fixed cost + variable cost

  $C = F + cQ$
- Revenue = price times quantity sold

  $R = pS$
- Quantity sold = Minimum{demand, quantity sold}

  $S = \min\{D, Q\}$
- Profit = Revenue — Cost

  $P = p^*\min\{D, Q\} - (F + cQ)$

Logic-Driven Modeling

Example 8.2 (continued) A Profit Model

- p = $40
- c = $24
- F = $400,000
- D = 50,000
- Q = 40,000

Compute:

- $R = p^*\min\{D,Q\}$
  $= 40(40{,}000) = 1{,}600{,}000$
- $C = F + cQ = 1{,}360{,}000$
- $= 400{,}000 + 24(40{,}000)$
- $P = R - C = 1{,}600{,}000 - 1{,}360{,}000 = \$240{,}000$

	A	B	C
1	Profit Model		
2			
3	Data		
4			
5	Unit Price	$40.00	
6	Unit Cost	$24.00	
7	Fixed Cost	$400,000.00	
8	Demand	50000	
9			
10			
11	Model		
12			
13	Unit Price	$40.00	
14	Quantity Sold	40000	
15	Revenue		$1,600,000.00
16			
17	Unit Cost	$24.00	
18	Quantity Produced	40000	
19	Variable Cost		$960,000.00
20	Fixed Cost		$400,000.00
21			
22	Profit		$240,000.00

Model8.2a

Logic-Driven Modeling

Example 8.2 (continued) A Profit Model

	A	B	C
1	Profit Model		
2			
3	Data		
4			
5	Unit Price	$40.00	
6	Unit Cost	$24.00	
7	Fixed Cost	$400,000.00	
8	Demand	50000	
9			
10			
11	Model		
12			
13	Unit Price	$40.00	
14	Quantity Sold	40000	
15	Revenue		$1,600,000.00
16			
17	Unit Cost	$24.00	
18	Quantity Produced	40000	
19	Variable Cost		$960,000.00
20	Fixed Cost		$400,000.00
21			
22	Profit		$240,000.00

Figure 8.2a

	A	B	C
1	Profit Model		
2			
3	Data		
4			
5	Unit Price	40	
6	Unit Cost	24	
7	Fixed Cost	400000	
8	Demand	50000	
9			
10			
11	Model		
12			
13	Unit Price	=B5	
14	Quantity Sold	=MIN(B8,B18)	
15	Revenue		=B13*B14
16			
17	Unit Cost	=B6	
18	Quantity Produced	40000	
19	Variable Cost		=B17*B18
20	Fixed Cost		=B7
21			
22	Profit		=C15-C19-C20

Figure 8.2b

Logic-Driven Modeling

Example 8.3 New-Product Development

- Moore Pharmaceuticals needs to decide whether to conduct clinical trials and seek FDA approval for a newly developed drug.

Estimated figures:

- R&D cost = $700 million
- Clinical trials cost = $150 million
- Market size = 2 million people
- Market size growth = 3% per year

Logic-Driven Modeling

Example 8.3 (continued) New-Product Development

Additional estimated figures

- Market share = 8%
- Market share growth = 20% per year (for 5 years)
- Revenue from a monthly prescription = $130
- Variable cost for a monthly prescription = $40
- Discount rate for net present value = 9%

Moore Pharmaceuticals wants to determine net present value for the next 5 years and to determine how long it will take to recover fixed costs.

Logic-Driven Modeling

Example 8.3 (continued) New-Product Development

	A	B	C	D	E	F
1	Moore Pharmaceuticals					
2						
3	Data					
4						
5	Market size	2000000				
6	Unit (monthly Rx) revenue	130				
7	Unit (monthly Rx) cost	40				
8	Discount rate	0.09				
9						
10	Project Costs					
11	R&D	700000000				
12	Clinical Trials	150000000				
13	Total Project Costs	=B11+B12				
14						
15	Model					
16						
17	Year 1		2	3	4	5
18	Market growth factor		0.03	0.03	0.03	0.03
19	Market size	=B5	=B19*(1+C18)	=C19*(1+D18)	=D19*(1+E18)	=E19*(1+F18)
20	Market share growth rate		0.2	0.2	0.2	0.2
21	Market share	0.08	=B21*(1+C20)	=C21*(1+D20)	=D21*(1+E20)	=E21*(1+F20)
22	Sales	=B19*B21	=C19*C21	=D19*D21	=E19*E21	=F19*F21
23						
24	Annual Revenue	=B22*$B$6*12	=C22*$B$6*12	=D22*$B$6*12	=E22*$B$6*12	=F22*$B$6*12
25	Annual Costs	=B22*$B$7*12	=C22*$B$7*12	=D22*$B$7*12	=E22*$B$7*12	=F22*$B$7*12
26	Profit	=B24-B25	=C24-C25	=D24-D25	=E24-E25	=F24-F25
27						
28	Cumulative Net Profit	=B26-B13	=B28+C26	=C28+D26	=D28+E26	=E28+F26
29						
30	Net Present Value	=NPV(B8,B26:F26)-B13				

Figure 8.3b

Logic-Driven Modeling

Example 8.3 (continued) New-Product Development

	A	B	C	D	E	F
1	Moore Pharmaceuticals					
2						
3	Data					
4						
5	Market size	2,000,000				
6	Unit (monthly Rx) revenue	$ 130.00				
7	Unit (monthly Rx) cost	$ 40.00				
8	Discount rate	9%				
9						
10	Project Costs					
11	R&D	$ 700,000,000				
12	Clinical Trials	$ 150,000,000				
13	Total Project Costs	$ 850,000,000				
14						
15	Model					
16						
17	Year	1	2	3	4	5
18	Market growth factor		3.00%	3.00%	3.00%	3.00%
19	Market size	2,000,000	2,060,000	2,121,800	2,185,454	2,251,018
20	Market share growth rate		20.00%	20.00%	20.00%	20.00%
21	Market share	8.00%	9.60%	11.52%	13.82%	16.59%
22	Sales	160,000	197,760	244,431	302,117	373,417
23						
24	Annual Revenue	$ 249,600,000	$ 308,505,600	$ 381,312,922	$ 471,302,771	$ 582,530,225
25	Annual Costs	$ 76,800,000	$ 94,924,800	$ 117,327,053	$ 145,016,237	$ 179,240,069
26	Profit	$ 172,800,000	$ 213,580,800	$ 263,985,869	$ 326,286,534	$ 403,290,156
27						
28	Cumulative Net Profit	$ (677,200,000)	$ (463,619,200)	$ (199,633,331)	$ 126,653,203	$ 529,943,358
29						
30	Net Present Value	$ 185,404,860				

NPV = $185 million

Profitable in 4th year

Figure 8.3a

Logic-Driven Modeling

Single-Period Purchase Decisions

- One-time purchase decisions often must be made in the face of uncertain demand.

Newsvendor Problem:

How many newspapers to purchase each day?

- C = cost to purchase a newspaper
- Q = number of newspapers the vendor purchases
- D = number of newspapers demanded
- R = revenue from selling a newspaper
- S = salvage value of unsold newspapers
- Net profit = $R(\min\{Q,D\}) + S(\max\{0,Q-D\}) - CQ$

Logic-Driven Modeling

Example 8.4

A Single-Period Purchase Decision Model

- Net profit $=18(\min\{Q,D\}) + 9(\max\{0,Q-D\}) - 12Q$

	A	B
1	Newsvendor Model	
2		
3	Data	
4		
5	Selling price	$ 18.00
6	Cost	$ 12.00
7	Discount price	$ 9.00
8		
9	Model	
10		
11	Demand	41
12	Purchase Quantity	44
13		
14	Quantity Sold	41
15	Surplus Quantity	3
16		
17	Profit	$ 237.00

	A	B
1	Newsvendor Model	
2		
3	Data	
4		
5	Selling price	18
6	Cost	12
7	Discount price	9
8		
9	Model	
10		
11	Demand	41
12	Purchase Quantity	44
13		
14	Quantity Sold	=MIN(B11,B12)
15	Surplus Quantity	=MAX(0,B12-B11)
16		
17	Profit	=B14*B5+B15*B7-B12*B6

Figure 8.4

Logic-Driven Modeling

Example 8.5 A Hotel Overbooking Model

- A popular resort hotel has 300 rooms.
- The room rate is $120 per night.
- Reservations can be cancelled by 6:00 p.m.
- Cost of overbooking is $100 per occurrence.

Determine net revenue on the rooms.

- $Q = 300$, $P = 120$, $C = 100$
- D = Reservations − Cancellations
- Net revenue $= P(\min\{300,D\}) - C(\max\{0,D-Q\})$
  $= 120(\min\{300,D\}) - 100(\max\{0,D-300\})$

Logic-Driven Modeling

Example 8.5 (continued)

A Hotel Overbooking Model

Net revenue = $120(\min\{300, D\}) - 100(\max\{0, D - 300\})$

	A	B
1	Hotel Overbooking Model	
2		
3	Data	
4		
5	Rooms available	300
6	Price	$120
7	Overbooking cost	$100
8		
9	Model	
10		
11	Reservation limit	300
12	Customer demand	290
13	Reservations made	290
14	Cancellations	15
15	Customer arrivals	275
16	Overbooked customers	0
17		
18	Net revenue	$33,000

	A	B
1	Hotel Overbooking Model	
2		
3	Data	
4		
5	Rooms available	300
6	Price	120
7	Overbooking cost	100
8		
9	Model	
10		
11	Reservation limit	300
12	Customer demand	290
13	Reservations made	=MIN(B11,B12)
14	Cancellations	15
15	Customer arrivals	=B13-B14
16	Overbooked customers	=MAX(0,B15-B5)
17		
18	Net revenue	=MIN(B15,B5)*B6-B16*B7

Figure 8.5

Logic-Driven Modeling

Example 8.6 A Retirement-Planning Model

- Start work at age 22, earning $50,000 per year.
- Expect a salary increase of 3% per year.
- Required to contribute 8% to retirement.
- Employer contributes 35% of that amount.
- Expect an annual return of 8% on the portfolio.

Determine the value of the retirement account when the employee is 50 years old.

Logic-Driven Modeling

Example 8.6 (continued) Retirement-Planning Model

- Salary = 1.03(previous year's salary)
- Employee contribution = 0.08(salary)
- Employer contribution = 0.35(employee contrib.)
- Value of account = 1.08(previous value) + employee contribution + employer contribution

	A	B	C	D	E
1	**Retirement Plan Model**				
2					
3	**Data**				
4					
5	**Retirement contribution (% of salary)**	0.08			
6	**Employer match**	0.35			
7	**Annual salary increase**	0.03			
8	**Annual return on investment**	0.08			
9					
10	**Model**				
11					
12			**Employee**	**Employer**	
13	**Age**	**Salary**	**Contribution**	**Contribution**	**Balance**
14	22	50000	=B14*$B$5	=$B$6*C14	=C14+D14
15	23	= B14*(1+$B$7)	=B15*$B$5	=$B$6*C15	=E14*(1+$B$8) + C15+D15
16	24	= B15*(1+$B$7)	=B16*$B$5	=$B$6*C16	=E15*(1+$B$8) + C16+D16
17	25	= B16*(1+$B$7)	=B17*$B$5	=$B$6*C17	=E16*(1+$B$8) + C17+D17

Figure 8.6a

Logic-Driven Modeling

Example 8.6 (continued) Retirement Planning Model

	A	B	C	D	E
1	**Retirement Plan Model**				
2					
3	**Data**				
4					
5	**Retirement contribution (% of salary)**	8%			
6	**Employer match**	35%			
7	**Annual salary increase**	3%			
8	**Annual return on investment**	8%			
9					
10	**Model**				
11					
12			**Employee**	**Employer**	
13	**Age**	**Salary**	**Contribution**	**Contribution**	**Balance**
14	22	$50,000	$4,000	$1,400	$5,400
15	23	$ 51,500	$4,120	$1,442	$11,394
16	24	$ 53,045	$4,244	$1,485	$18,034
17	25	$ 54,636	$4,371	$1,530	$25,378
18	26	$ 56,275	$4,502	$1,576	$33,486
19	27	$ 57,964	$4,637	$1,623	$42,425
20	28	$ 59,703	$4,776	$1,672	$52,267
21	29	$ 61,494	$4,919	$1,722	$63,089
22	30	$ 63,339	$5,067	$1,773	$74,977
23	31	$ 65,239	$5,219	$1,827	$88,021
24	32	$ 67,196	$5,376	$1,881	$102,320
25	33	$ 69,212	$5,537	$1,938	$117,980
26	34	$ 71,288	$5,703	$1,996	$135,118
27	35	$ 73,427	$5,874	$2,056	$153,857
28	36	$ 75,629	$6,050	$2,118	$174,334
29	37	$ 77,898	$6,232	$2,181	$196,694
30	38	$ 80,235	$6,419	$2,247	$221,094
31	39	$ 82,642	$6,611	$2,314	$247,707
32	40	$ 85,122	$6,810	$2,383	$276,717
33	41	$ 87,675	$7,014	$2,455	$308,323
34	42	$ 90,306	$7,224	$2,529	$342,742
35	43	$ 93,015	$7,441	$2,604	$380,207
36	44	$ 95,805	$7,664	$2,683	$420,971
37	45	$ 98,679	$7,894	$2,763	$465,306
38	46	$ 101,640	$8,131	$2,846	$513,507
39	47	$ 104,689	$8,375	$2,931	$565,894
40	48	$ 107,830	$8,626	$3,019	$622,811
41	49	$ 111,064	$8,885	$3,110	$684,631
42	50	$ 114,396	$9,152	$3,203	$751,757

Value at 22 years old = $5,400

Value at 50 years old = $751,757

Figure 8.6b

Data-Driven Modeling

Example 8.7
Modeling Retail Markdown Pricing Decisions

- In the spring, a department store introduces a new line of bathing suits that sells for $70.
- The store purchases 1000 of these bathing suits.
- During the prime selling season, the store sells an average of 7 units per day at full price (40 days).
- On 10 sale days, the price is discounted 30% and sales increase to 32.2 units per day.
- Around July 4th, the price is marked down 70% to sell off remaining inventory.
- Determine total revenue from the bathing suits.

Data-Driven Modeling

Example 8.7 (continued)
Modeling Retail Markdown Pricing Decisions

Assume a linear trend model between sales and price:

daily sales = $a - b$(price)

$7 = a - b(70)$

$32.2 = a - b(49)$

Daily sales = 91 – 1.2(price)

	A	B	C
1	**Markdown Pricing Model**		
2			
3	**Data**		
4	**Retail price**	70	
5	**Inventory**	1000	
6	**Selling season (days)**	50	
7	**Days at full retail**	40	
8	**Intermediate markdown**	0.3	
9	**Clearance markdown**	0.7	
10	**Demand function**		
11	**a**	91	
12	**b**	1.2	
13			
14	**Model**		
15			
16	**Full Retail Sales**		
17	**Retail price**	=B4	
18	**Daily sales**	=B11-B12*B17	
19	**Days at retail price**	=B7	
20	**Units sold at retail**	=B18*B19	
21	**Retail revenue**	=B20*B17	
22	**Discount Sales**		
23	**Discount**	=B8	
24	**Discount price**	=B17*(1-B23)	
25	**Daily sales**	=B11 - B12*B24	
26	**Unit sold**	=MIN(B25*(B6-B19),B5-B20)	
27	**Discount revenue**	=B26*B24	
28	**Clearance Sales**		
29	**Clearance price**	=B4*(1-B9)	
30	**Units sold at clearance**	=MAX(0,B5-B20-B26)	
31	**Clearance revenue**	=B29*B30	
32			
33	**Total revenue**	=C21+C27+C31	

Figure 8.7

Data-Driven Modeling

Example 8.7 (continued)

Revenue from full retail sales
= units sold * days * price
= (7)*(40)*(70)
= $19,600
Revenue from sale weekends
= (32.2)*(10)*(49)
= $15,778
Revenue from clearance sales
= leftovers * price
= (1000−7(40) − 32.2(10))*(21)
= (398)(21)
= $8,358

	A	B	C
1	Markdown Pricing Model		
2			
3	Data		
4	Retail price	70	
5	Inventory	1000	
6	Selling season (days)	50	
7	Days at full retail	40	
8	Intermediate markdown	0.3	
9	Clearance markdown	0.7	
10	Demand function		
11	a	91	
12	b	1.2	
13			
14	Model		
15			
16	Full Retail Sales		
17	Retail price	=B4	
18	Daily sales	=B11-B12*B17	
19	Days at retail price	=B7	
20	Units sold at retail	=B18*B19	
21	Retail revenue	=B20*B17	
22	Discount Sales		
23	Discount	=B8	
24	Discount price	=B17*(1-B23)	
25	Daily sales	=B11 - B12*B24	
26	Unit sold	=MIN(B25*(B6-B19),B6-B20)	
27	Discount revenue	=B26*B24	
28	Clearance Sales		
29	Clearance price	=B4*(1-B9)	
30	Units sold at clearance	=MAX(0,B5-B20-B26)	
31	Clearance revenue	=B29*B30	
32			
33	Total revenue	=C21+C27+C31	

Figure 8.7

Data-Driven Modeling

Example 8.7 (continued)
Modeling Retail Markdown Pricing Decisions

	A	B	C
1	Markdown Pricing Model		
2			
3	Data		
4	Retail price	70	
5	Inventory	1000	
6	Selling season (days)	50	
7	Days at full retail	40	
8	Intermediate markdown	0.3	
9	Clearance markdown	0.7	
10	Demand function		
11	a	91	
12	b	1.2	
13			
14	Model		
15			
16	Full Retail Sales		
17	Retail price	=B4	
18	Daily sales	=B11-B12*B17	
19	Days at retail price	=B7	
20	Units sold at retail	=B18*B19	
21	Retail revenue	=B20*B17	
22	Discount Sales		
23	Discount	=B8	
24	Discount price	=B17*(1-B23)	
25	Daily sales	=B11 - B12*B24	
26	Unit sold	=MIN(B25*(B6-B19),B6-B20)	
27	Discount revenue	=B26*B24	
28	Clearance Sales		
29	Clearance price	=B4*(1-B9)	
30	Units sold at clearance	=MAX(0,B5-B20-B26)	
31	Clearance revenue	=B29*B30	
32			
33	Total revenue	=C21+C27+C31	

	A	B	C
1	Markdown Pricing Model		
2			
3	Data		
4	Retail price	$70.00	
5	Inventory	1000	
6	Selling season (days)	50	
7	Days at full retail	40	
8	Intermediate markdown	30%	
9	Clearance markdown	70%	
10	Demand function		
11	a	91	
12	b	1.2	
13			
14	Model		
15			
16	Full Retail Sales		
17	Retail price	$70.00	
18	Daily sales	7.00	
19	Days at retail price	40	
20	Units sold at retail	280	
21	Retail revenue	$19,600.00	
22	Discount Sales		
23	Discount	30%	
24	Discount price	$49.00	
25	Daily sales	32.20	
26	Unit sold	322	
27	Discount revenue	$15,778.00	
28	Clearance Sales		
29	Clearance price	$21.00	
30	Units sold at clearance	398	
31	Clearance revenue	$8,358.00	
32			
33	Total revenue	$43,736.00	

Total revenue
= $43,736

Figure 8.7

Data-Driven Modeling

Modeling Relationships and Trends in Data

- Create charts to better understand data sets.
- For cross-sectional data, use a scatter chart.
- For time series data, use a line chart.
- Consider using mathematical functions to model relationships.

Data-Driven Modeling

Excel *Trendline* tool

Click on a chart
- *Chart tools*
- *Layout*
- *Trendline*

Choose a Trendline. Choose whether to display equation and *R*-squared.

R-squared values closer to 1 indicate better fit of the Trendline to the data.



Figure 8.8

Data-Driven Modeling

Example 8.8 Modeling a Price-Demand Function

Linear demand function:
Sales = -9.5116(price) + 20512

	A	B
1	Price-Sales Data	
2		
3	Price	Demand
4	$50.00	19964.085
5	$60.00	19706.855
6	$70.00	20240.83
7	$80.00	19698.807
8	$90.00	20095.808
9	$100.00	19390.994
10	$110.00	19430.071
11	$120.00	19273.695
12	$130.00	18716.381
13	$140.00	18925.364
14	$150.00	19484.776
15	$160.00	18934.879
16	$170.00	18915.765
17	$180.00	18893.373
18	$190.00	18961.622
19	$200.00	18443.294
20	$210.00	18811.981
21	$220.00	18561.917
22	$230.00	18158.619
23	$240.00	18412.56
24	$250.00	17771.387



Figure 8.9

Data-Driven Modeling

Example 8.9 Predicting Crude Oil Prices

- Line chart of historical crude oil prices



Figure 8.10

Data-Driven Modeling

Example 8.9 (continued) Predicting Crude Oil Prices

- Excel's *Trendline* tool is used to fit various functions to the data.

Logarithmic $y = 13 \ln(x) + 39$ $R^2 = 0.382$

Power $y = 45.96x^{0.0169}$ $R^2 = 0.397$

Exponential $y = 50.5e^{0.021x}$ $R^2 = 0.664$

Polynomial 2° $y = 0.13x^2 - 2.4x + 68$ $R^2 = 0.905$

Polynomial 3° $y = 0.005x^3 - 0.111x^2 + 0.648x + 59.5$ $R^2 = 0.928$ *

Data-Driven Modeling

Example 8.9 (continued) Predicting Crude Oil Prices

- Third Order Polynomial Trendline fit to the data



Figure 8.11

Analyzing Uncertainty and Model Assumptions

What-If Analysis

- Spreadsheet models allow you to easily evaluate what-if questions.
- How do changes in model inputs (that reflect key assumptions) affect model outputs?
- Systematic approaches to what-if analysis make the process easier and more useful.

Analyzing Uncertainty and Model Assumptions

Data Tables

- Data Tables summarize the impact of one or two inputs on a specified output.
- Excel data table types:
  - One-way data tables – for one input variable
  - Two-way data table – for two input variables

To construct a data table:

- *Data*
- *What-If Analysis*
- *Data Table*



Figure 8.14

Analyzing Uncertainty and Model Assumptions

Example 8.11

A One-Way Data Table for Uncertain Demand

Create a column of demand values (column E).
Enter =C22 in cell F3 (to reference the output cell).
Highlight the range E3:F11.
Choose *Data Table*.
Enter B8 for *Column input cell.* *(tells Excel that column E is demand values)*

	A	B	C	D	E	F
1	**Profit Model**					
2						
3	**Data**				**Demand**	
4					25000	*Data Table tool*
5	**Unit Price**	$40.00			30000	*computes*
6	**Unit Cost**	$24.00			35000	*these*
7	**Fixed Cost**	$400,000.00			40000	*values*
8	**Demand**	50000			45000	
9					50000	
10					55000	
11	**Model**				60000	$ 240,000.00
12						
13	**Unit Price**	$40.00				
14	**Quantity Sold**	40000				
15	**Revenue**		$1,600,000.00			
16						
17	**Unit Cost**	$24.00				
18	**Quantity Produced**	40000				
19	**Variable Cost**		$960,000.00			
20	**Fixed Cost**		$400,000.00			
21						
22	**Profit**		$240,000.00			

Figure 8.15a



Figure 8.14

Analyzing Uncertainty and Model Assumptions

Example 8.11 (continued)

A One-Way Data Table for Uncertain Demand

The Data Table tool computes the profit values in column F (below $240,000).

	A	B	C	D	E	F
1	**Profit Model**					
2						
3	**Data**				**Demand**	$240,000.00
4					25000	$ (360,000.00)
5	**Unit Price**	$40.00			30000	$ (160,000.00)
6	**Unit Cost**	$24.00			35000	$ 40,000.00
7	**Fixed Cost**	$400,000.00			40000	$ 240,000.00
8	**Demand**	50000			45000	$ 240,000.00
9					50000	$ 240,000.00
10					55000	$ 240,000.00
11	**Model**				60000	$ 240,000.00
12						
13	**Unit Price**	$40.00				
14	**Quantity Sold**	40000				
15	**Revenue**		$1,600,000.00			
16						
17	**Unit Cost**	$24.00				
18	**Quantity Produced**	40000				
19	**Variable Cost**		$960,000.00			
20	**Fixed Cost**		$400,000.00			
21						
22	**Profit**		$240,000.00			

Figure 8.15b

Analyzing Uncertainty and Model Assumptions

Example 8.12

One-Way Data Tables with Multiple Outputs

- Create a second output, revenue.

Enter =C15 in cell G3.
Highlight E3:G11.
Choose *Data Table*
Proceed as in the previous example.
Excel computes the revenues values.

	A	B	C	D	E	F	G
1	Profit Model						
2							
3	Data					Profit	Revenue
4					Demand	$240,000.00	$1,600,000.00
5	Unit Price	$40.00			25000	$ (360,000.00)	$ 1,000,000.00
6	Unit Cost	$24.00			30000	$ (160,000.00)	$ 1,200,000.00
7	Fixed Cost	$400,000.00			35000	$ 40,000.00	$ 1,400,000.00
8	Demand	50000			40000	$ 240,000.00	$ 1,600,000.00
9					45000	$ 240,000.00	$ 1,600,000.00
10					50000	$ 240,000.00	$ 1,600,000.00
11	Model				55000	$ 240,000.00	$ 1,600,000.00
12					60000	$ 240,000.00	$ 1,600,000.00
13	Unit Price	$40.00					
14	Quantity Sold	40000					
15	Revenue		$1,600,000.00				
16							
17	Unit Cost	$24.00					
18	Quantity Produced	40000					
19	Variable Cost		$960,000.00				
20	Fixed Cost		$400,000.00				
21							
22	Profit		$240,000.00				

Figure 8.15

Analyzing Uncertainty and Model Assumptions

Example 8.13

A Two-Way Data Table for the Profit Model

- Evaluate the impact of both unit price and unit cost

Create a column of unit prices (F5:F15).
Create a row of unit costs (G4:J4).
Enter =C22 in cell F4.
Select F4:J15.
Choose *Data Table*.

	A	B	C	D	E	F	G	H	I	J
1	Profit Model									
2										
3	Data						Unit Cost			
4					Unit	$240,000.00	$22.00	$23.00	$24.00	$25.00
5	Unit Price	$40.00			Price	$35.00				
6	Unit Cost	$24.00				$36.00				
7	Fixed Cost	$400,000.00				$37.00				
8	Demand	50000				$38.00				
9						$39.00				
10						$40.00				
11	Model					$41.00				
12						$42.00				
13	Unit Price	$40.00				$43.00				
14	Quantity Sold	40000				$44.00				
15	Revenue		$1,600,000.00			$45.00				
16										
17	Unit Cost	$24.00								
18	Quantity Produced	40000								
19	Variable Cost		$960,000.00							
20	Fixed Cost		$400,000.00							
21										
22	Profit		$240,000.00							

Data Table tool computes these cell values.

Figure 8.17a

Enter B6 for *Row input cell*.
Enter B5 for *Column input cell*.

Analyzing Uncertainty and Model Assumptions

Example 8.13 (continued)

A Two-Way Data Table for the Profit Model

	A	B	C	D	E	F	G	H	I	J
1	**Profit Model**									
2										
3	**Data**						**Unit Cost**			
4					**Unit**	**$240,000.00**	**$22.00**	**$23.00**	**$24.00**	**$25.00**
5	**Unit Price**	$40.00			**Price**	$35.00	$120,000.00	$80,000.00	$40,000.00	$0.00
6	**Unit Cost**	$24.00				$36.00	$160,000.00	$120,000.00	$80,000.00	$40,000.00
7	**Fixed Cost**	$400,000.00				$37.00	$200,000.00	$160,000.00	$120,000.00	$80,000.00
8	**Demand**	50000				$38.00	$240,000.00	$200,000.00	$160,000.00	$120,000.00
9						$39.00	$280,000.00	$240,000.00	$200,000.00	$160,000.00
10						$40.00	$320,000.00	$280,000.00	$240,000.00	$200,000.00
11	**Model**					$41.00	$360,000.00	$320,000.00	$280,000.00	$240,000.00
12						$42.00	$400,000.00	$360,000.00	$320,000.00	$280,000.00
13	**Unit Price**	$40.00				$43.00	$440,000.00	$400,000.00	$360,000.00	$320,000.00
14	**Quantity Sold**	40000				$44.00	$480,000.00	$440,000.00	$400,000.00	$360,000.00
15	**Revenue**		$1,600,000.00			$45.00	$520,000.00	$480,000.00	$440,000.00	$400,000.00
16										
17	**Unit Cost**	$24.00								
18	**Quantity Produced**	40000								
19	**Variable Cost**		$960,000.00							
20	**Fixed Cost**		$400,000.00							
21										
22	**Profit**		$240,000.00							

Figure 8.17b

Analyzing Uncertainty and Model Assumptions

Goal Seek

Goal Seek allows you to alter the data used in a formula in order to find out what the results will be.

- *Set cell* contains the formula that will return the result you're seeking.
- *To value* is the target value you want the formula to return.
- *By changing cell* is the location of the input value that Excel can change to reach the target.



Figure 8.21

Analyzing Uncertainty and Model Assumptions

Example 8.15 Finding the Breakeven Point in the Outsourcing Model (using *Goal Seek*)

- Find the value of demand at which manufacturing cost equals purchased cost
- *Set cell*: B19
- *To value*: 0
- *By changing cell*: B12.



Figure 8.21

	A	B
1	Outsourcing Decision Model	
2		
3	Data	
4		
5	Manufactured in-house	
6	Fixed cost	$50,000
7	Unit variable cost	$125
8		
9	Purchased from supplier	
10	Unit cost	$175
11		
12	Production volume	1000
13		
14	Model	
15		
16	Total manufacturing cost	$175,000
17	Total purchased cost	$175,000
18		
19	Cost difference	$0
20	Decision	Manufacture

Figure 8.22

The breakeven volume is 1000 units.

Model Analysis Using *Risk Solver Platform*

Tornado Chart

- Shows the impact that variation in a model input has on some output while holding all other inputs constant.
- Shows which inputs are the least and most influential on the output.
- Helps you select the inputs that you would want to further analyze.

Model Analysis Using *Risk Solver Platform*

Example 8.17

Creating a Tornado Chart in *Risk Solver Platform*

Profit Model

Select cell C22.
Parameters
Identify

A 10% change in unit price (B5)
affects profit the most.
Next is unit cost (B6).



Figure 8.28